

Gramática LL1 da Linguagem GF
Alunos: Filipe Falcão e Gabriel Nunes
Período: 2018.2

Obs.: (i) não-terminais cujos nomes terminam com os termos *fat* ou *r* indicam que se fez necessário realizar uma fatoração ou eliminar uma recursão (à esquerda) para manter a gramática no padrão de um analisador LL(1); e (ii) a gramática original se encontra disponível em <https://bit.ly/2PyKoG>.

PROGRAM = FUNCTIONS MAINFUNC

FUNCTIONS = 'funDecl' 'id' PARAMS RETURNTYPE BLOCK FUNCTIONS
FUNCTIONS = ϵ

MAINFUNC = 'main' 'paramBeg' 'paramEnd' TYPE BLOCK

PARAMS = 'paramBeg' PARAMSfat
PARAMSfat = 'paramEnd'
PARAMSfat = LISTPARAMS 'paramEnd'

LISTPARAMS = TYPE NAME LISTPARAMSr
LISTPARAMSr = 'commaSep' LISTPARAMS
LISTPARAMSr = ϵ

TYPE = 'typeInt'
TYPE = 'typeFloat'
TYPE = 'typeBool'
TYPE = 'typeChar'
TYPE = 'typeString'
TYPE = 'typeEmpty'
TYPE = 'typeNull'

RETURNTYPE = TYPE RETURNTYPEfat
RETURNTYPEfat = 'arrayBeg' 'arrayEnd'
RETURNTYPEfat = ϵ

NAME = 'id' NAMEfat
NAMEfat = 'arrayBeg' EXPR 'arrayEnd'
NAMEfat = ϵ

NAMEfat = 'arrayBeg' EXPR 'arrayEnd'

NAMEfat2 = EXPR 'arrayEnd'

NAMEfat2 = 'arrayEnd'

BLOCK = 'blockBeg' CMDS 'blockEnd'

CMDS = CMD CMDS

CMDS = ϵ

CMD = DECL 'endLine'

CMD = id CMDfat

CMD = PRINT 'endLine'

CMD = READ 'endLine'

CMD = IFELSE

CMD = WHILE

CMD = REPEAT

CMD = RETURN 'endLine'

CMDfat = 'arrayBeg' EXPR 'arrayEnd' ASSG 'endLine'

CMDfat = ASSG 'endLine'

CMDfat = FUNCCALL 'endLine'

CMDfat = ϵ

DECL = 'varDecl' TYPE NAME

ASSG = 'opAssign' VALUE

VALUE = ARRAY

VALUE = EXPR

ARRAY = 'arrayBeg' ELEMENTS 'arrayEnd'

ELEMENTS = EXPR ELEMENTSr

ELEMENTSr = 'commaSep' ELEMENTS

ELEMENTSr = ϵ

CONSTANT = 'constNumInt'

CONSTANT = 'constNumFloat'

CONSTANT = 'constBool'

CONSTANT = 'constChar'

CONSTANT = 'constString'

FUNCCALL = 'paramBeg' LISTPARAMSCALL 'paramEnd'

LISTPARAMSCALL = PARAMITEM LISTPARAMSCALLr
LISTPARAMSCALLr = 'commaSep' LISTPARAMSCALL
LISTPARAMSCALLr = ϵ

PARAMITEM = CONSTANT
PARAMITEM = NAME PARAMITEMfat
PARAMITEM = ϵ
PARAMITEMfat = FUNCCALL
PARAMITEMfat = ϵ

PRINT = 'cmdPrint' 'paramBeg' MESSAGE 'paramEnd'
READ = 'cmdRead' 'paramBeg' NAME 'paramEnd'

MESSAGE = 'constString' MESSAGEr
MESSAGE = NAME MESSAGEr
MESSAGEr = 'opConcat' MESSAGE
MESSAGEr = ϵ

IFELSE = IF ELIF ELSE

IF = 'cmdIf' 'paramBeg' EXPR 'paramEnd' BLOCK

ELIF = 'cmdElif' 'paramBeg' EXPR 'paramEnd' BLOCK ELIF
ELIF = ϵ

ELSE = 'cmdElse' BLOCK
ELSE = ϵ

WHILE = 'cmdWhile' 'paramBeg' EXPR 'paramEnd' BLOCK

REPEAT = 'cmdRepeat' 'id' 'cmdIn' 'paramBeg' EXPR 'commaSep'
EXPR REPEATfat 'paramEnd' BLOCK
REPEATfat = 'commaSep' EXPR
REPEATfat = ϵ

RETURN = 'cmdReturn' RETURNfat
RETURNfat = CONSTANT
RETURNfat = NAME

EXPR = EXPROR
EXPROR = EXPRAND EXPRORr

EXPRORr = 'opOr' EXPRAND EXPRORr
EXPRORr = ϵ

EXPRAND = EXPREQUALS EXPRANDr
EXPRANDr = 'opAnd' EXPREQUALS EXPRANDr
EXPRANDr = ϵ

EXPREQUALS = EXPRREL EXPREQUALSr
EXPREQUALSr = 'opEquals' EXPRREL EXPREQUALSr
EXPREQUALSr = ϵ

EXPRREL = EXPRADD EXPRRELr
EXPRRELr = 'opRel' EXPRADD EXPRRELr
EXPRRELr = ϵ

EXPRADD = EXPRAMULT EXPRADDr
EXPRADDr = 'opAditiv' EXPRAMULT EXPRADDr
EXPRADDr = ϵ

EXPRAMULT = EXPRUN EXPRAMULTr
EXPRAMULTr = 'opMult' EXPRUN EXPRAMULTr
EXPRAMULTr = ϵ

EXPRUN = 'opUnaryNeg' EXPRUN
EXPRUN = EXPRPARAM

EXPRPARAM = 'paramBeg' EXPR 'paramEnd'
EXPRPARAM = EXPRVALUES

EXPRVALUES = 'constNumInt'
EXPRVALUES = 'constNumFloat'
EXPRVALUES = 'constBool'
EXPRVALUES = 'constChar'
EXPRVALUES = 'constString'
EXPRVALUES = 'id' EXPRVALUESfat
EXPRVALUESfat = 'arrayBeg' EXPR 'arrayEnd'
EXPRVALUESfat = FUNCCALL
EXPRVALUESfat = ϵ