Gramática LL1 da Linguagem GF
Alunos: Filipe Falcão e Gabriel Nunes
Período: 2018.2

```
PROGRAM = FUNCTIONS MAINFUNC

FUNCTIONS = ‘funDecl’ ‘id’ PARAMS RETURNTYPE BLOCK FUNCTIONS
FUNCTIONS = ε

MAINFUNC = ‘main’ ‘paramBeg’ ‘paramEnd’ TYPE BLOCK

PARAMS = ‘paramBeg’ PARAMSFAT
PARAMSFAT = ‘paramEnd’
PARAMSFAT = LISTPARAMS ‘paramEnd’

LISTPARAMS = TYPE NAME LISTPARAMSFAT
LISTPARAMSFAT = ‘commaSep’ LISTPARAMS
LISTPARAMSFAT = ε

TYPE = ‘typeInt’
TYPE = ‘typeFloat’
TYPE = ‘typeBool’
TYPE = ‘typeChar’
TYPE = ‘typeString’
TYPE = ‘typeEmpty’
TYPE = ‘typeNull’

RETURNTYPE = TYPE RETURNTYPEFAT
RETURNTYPEFAT = ‘arrayBeg’ ‘arrayEnd’
RETURNTYPEFAT = ε

NAME = ‘id’ NAMEFAT
NAMEFAT = ‘arrayBeg’ EXPR ‘arrayEnd’
NAMEFAT = ε

BLOCK = ‘blockBeg’ CMDS ‘blockEnd’

CMDS = CMD CMDS
CMDS = ε
```

```
CMD = DECL ‘endLine’
CMD = id CMDFAT
CMD = PRINT ‘endLine’
CMD = READ ‘endLine’
CMD = IFELSE
CMD = WHILE
CMD = REPEAT
CMD = RETURN ‘endLine’

CMDFAT = ‘arrayBeg’ EXPR ‘arrayEnd’ ASSG ‘endLine’
CMDFAT = ASSG ‘endLine’
CMDFAT = FUNCCALL ‘endLine’
CMDFAT = ε

DECL = ‘varDecl’ TYPE NAME

ASSG = ‘opAssign’ VALUE

VALUE = ARRAY
VALUE = EXPR

ARRAY = ‘arrayBeg’ ARRAYFAT
ARRAYFAT = ELEMENTS ‘arrayEnd’

ELEMENTS = EXPR ELEMENTSFAT
ELEMENTSFAT = ‘commaSep’ ELEMENTS
ELEMENTSFAT = ε

CONSTANT = ‘constNumInt’
CONSTANT = ‘constNumFloat’
CONSTANT = ‘constBool’
CONSTANT = ‘constChar’
CONSTANT = ‘constString’

FUNCCALL = ‘paramBeg’ LISTPARAMSCALL ‘paramEnd’

LISTPARAMSCALL = PARAMITEM LISTPARAMSCALLFAT
LISTPARAMSCALLFAT = ‘commaSep’ LISTPARAMSCALL
LISTPARAMSCALLFAT = ε

PARAMITEM = CONSTANT
PARAMITEM = NAME PARAMITEMFAT
```

```
PARAMITEMFAT = FUNCCALL
PARAMITEMFAT = ε

PRINT = ‘cmdPrint’ ‘paramBeg’ MESSAGE ‘paramEnd’

MESSAGE = ‘constString’ MESSAGEFAT
MESSAGE = NAME MESSAGEFAT
MESSAGEFAT = ‘opConcat’ MESSAGE
MESSAGEFAT = ε

READ = ‘cmdRead’ ‘paramBeg’ NAME ‘paramEnd’

IFELSE = IF ELIF ELSE

IF = ‘cmdIf’ ‘paramBeg’ EXPRLOGIC ‘paramEnd’ BLOCK

ELIF = ‘cmdElif’ ‘paramBeg’ EXPRLOGIC ‘paramEnd’ BLOCK ELIF
ELIF = ε

ELSE = ‘cmdElse’ BLOCK
ELSE = ε

WHILE = ‘cmdWhile’ ‘paramBeg’ EXPRLOGIC ‘paramEnd’ BLOCK

REPEAT = ‘cmdRepeat’ REPEATPARAM ‘cmdIn’ ‘paramBeg’
REPEATPARAM ‘commaSep’ REPEATPARAM ‘commaSep’ REPEATPARAM
‘paramEnd’ BLOCK

REPEATPARAM = ‘id’
REPEATPARAM = ‘constNumInt’

RETURN = ‘cmdReturn’ RETURNFAT
RETURNFAT = CONSTANT
RETURNFAT = NAME

EXPR = EXPRLOGIC EXPR
EXPR = ε

EXPRLOGIC = EXPROR EXPRLOGICFAT
EXPRLOGICFAT = ‘arrayBeg’ EXPR ‘arrayEnd’ EXPRRELFAT
EXPRLOGICFAT = ε
```

```
EXPROR = EXPRAND EXPRORFAT
EXPRORFAT = ‘opOr’ EXPRAND
EXPRORFAT = ε

EXPRAND = EXPREQUALS EXPRANDFAT
EXPRANDFAT = ‘opAnd’ EXPREQUALS
EXPRANDFAT = ε

EXPREQUALS = EXPRREL EXPREQUALSFAT
EXPREQUALSFAT = ‘opEquals’ EXPRREL
EXPREQUALSFAT = ε

EXPRREL = EXPRADD EXPRRELFAT
EXPRRELFAT = ‘opRel’ EXPRADD
EXPRRELFAT = ε

EXPRADD = EXPRAMULT EXPRADDFAT
EXPRADDFAT = ‘opAditiv’ EXPRAMULT
EXPRADDFAT = ε

EXPRAMULT = EXPRUN EXPRAMULTFAT
EXPRAMULTFAT = ‘opMult’ EXPRUN
EXPRAMULTFAT = ε

EXPRUN = ‘opUnaryNeg’ EXPRUN
EXPRUN = EXPRPARAM

EXPRPARAM = ‘paramBeg’ EXPRLOGIC ‘paramEnd’
EXPRPARAM = EXPRVALUES

EXPRVALUES = ‘id’
EXPRVALUES = ‘constNumInt’
EXPRVALUES = ‘constNumFloat’
EXPRVALUES = ‘constBool’
EXPRVALUES = ‘constChar’
EXPRVALUES = ‘constString’
```